I.O.S.

AN ALGORITHM
TO COMPUTE THE GEOID SURFACE

BY

M. AMIN

REPORT NO. 168

1983

NATURAL ENVIRONMENT RESEARCH COUNCIL

INSTITUTE OF
OCEANOGRAPHIC
SCIENCES

INSTITUTE OF OCEANOGRAPHIC SCIENCES

Wormley, Godalming,
Surrey, GU8 5UB.
(0428 - 79 - 4141)

(Director: Dr. A.S. Laughton FRS)

Bidston Observatory,
Birkenhead,
Merseyside, L43 7RA.
(051 - 653 - 8633)

(Assistant Director: Dr. D.E. Cartwright)

Crossway,
Taunton,
Somerset, TA1 2DW.
(0823 - 86211)

(Assistant Director: M.J. Tucker)

When citing this document in a bibliography the reference should be given as

AMIN, M. 1983 An algorithm to compute the geoid surface. *Institute of Oceanographic Sciences, Report*, No. 168, 18pp.

INSTITUTE OF OCEANOGRAPHIC SCIENCES

BIDSTON

An algorithm to compute the geoid surface

by

M. Amin

I.O.S. Report No. 168

1983

Abstract

A new algorithm to generate the geoid surface is presented. It can use tesseral harmonic coefficients from any model such as GEM 10B or GEM 10C. The Associated Legendre functions involved in the computation are evaluated by simple recurrence relationships which are convergent and fast, thus the efficiency of reducing altimeter data to extract tides and surface geostrophic currents is increased.

Introduction

The precision of measuring range by altimeter has increased from ± 0.6m in the case of GEOS-3 and SKYLAB to ± 0.06m in case of SEASAT. This improvement has established satellite altimetry as a powerful tool in the field of global oceanography; in studying the spatial and temporal variations of ocean circulation,tides and waves. The accuracy of the altimeter height cannot serve a useful purpose unless supported by equally precise measurements of its orbital elevations, and some purposes also require a detailed knowledge of the gravimetric geoid (Wunsch & Gaposchkin, 1980). It is the purpose of this work to develop an algorithm to compute the low-wave number part of the geoid from the tesseral harmonic coefficients of any chosen model, such as GEM 10B or GEM 10C.

Mathematical backgound

The Earth's potential at a point rotating with it and situated a distance r from the centre, geocentric co-latitude θ and east longitude λ is given by

$$U(r,\theta,\lambda)= \frac{1}{2} w^2 r^2 \sin^2\theta + \frac{GM}{r}\left[1 + \sum_{\ell=2}^{\infty} \sum_{m=0}^{\ell} \left(\frac{a}{r}\right)^{\ell} P_{\ell m}(\cos\theta)\ \{\bar{C}_{\ell m}\cos m\lambda + \bar{S}_{\ell m}\sin nm\lambda\} N_{\ell m}\right] \qquad (1)$$

where w is the sidereal angular velocity of rotation of the Earth (72.92115×10^{-6} radians/s),

G is the gravitational constant,

M is the mass of the Earth ($GM = 398600.44\ km^3\ s^{-2}$),

a is the mean equatorial radius (6378.138 km),

$\bar{C}_m, \bar{S}_m$ are the normalised tesseral harmonic coefficients,

$P_{\ell m}(\cos\theta)$ is the Associated Lengendre function of degree ℓ and order m,

$N_{\ell m}$ is the normalising factor defined as

$$N^2_{\ell m} = (2-\delta_{0m})(2\ell - 1)\ \frac{(n-m)!}{(n+m)!} \qquad (2)$$

and where $\delta_{0m} = 1$ when m = 0, otherwise $\delta_{0m}=0$.

$\bar{P}_{\ell m} = P_{\ell m} N_{\ell m}$ is the fully normalised Associated Legengre function.

The potential of the Earth's gravity field, as given in equation (1), is due to two forces: centrifugal force represented by the first term and the gravitational force represented by the second term. For mathematical convenience equation (1) can be reduced to a non-dimensional form by dividing it by $\frac{GM}{a}$ to give

$$V(\rho,\theta,\lambda) = \alpha \rho^2 \sin^2\theta + \rho^{-1}\left[1+\sum_{\ell=2}^{\infty}\sum_{m=0}^{\ell} \rho^{-\ell}\bar{P}_{\ell m}(\cos\theta)\{\bar{C}_{\ell m}\cos m\lambda+\bar{S}_{\ell m}\sin m\lambda\}\right] \quad (3)$$

where $V = Ua/GM$ (4)

$$\rho = r/a \quad (5)$$

$$\alpha = w^2 a^3/2GM = 0.00173070 \quad (6)$$

The mean potential V_0 at the ellipsoid's equator (θ=90 and ρ= 1) is given by

$$V_0 = 1 + \alpha + \sum_{\ell=2}^{\infty} \bar{P}_{\ell 0}(0)\, \bar{C}_{\ell 0} \quad (7)$$

The geoid is the surface on which the potential of the Earth is strictly constant, chosen arbitrarily to be V_c . Consider the potential at a general point on the the mean ellipsoid as a reference surface. To compute the geoid we require the anomaly δr in r which would account for the potential difference ($V - V_c$). That is, for a given geographical latitude ϕ,

$$\delta r(\phi,\lambda) = \delta r(\rho,\theta,\lambda) = -a\,(V-V_0)/(\partial V/\partial\rho) \quad (8)$$

where $\partial V/\partial\rho = 2\alpha\rho\sin^2\theta - \rho^{-2}\left[1+\frac{3}{\rho^2}\bar{P}_{20}(\cos\theta)\,\bar{C}_{20}\right]$ (9)

The approximation (9) can be justified because the contribution of high degree and high order terms to $\partial V/\partial\rho$ is less than 0.001 per cent. This will have negligible effect on the accuracy of δr . The relation between geographical latitude ϕ and the geocentric θ and ρ can be computed from the following equations (Torge (1980),p.49):

$$\rho\sin\theta = \cos\phi/[\cos^2\phi + (1-f)^2\sin^2\phi] \tag{10a}$$

$$\rho\cos\theta = (1-f)^2 \sin\phi/[\cos^2\phi + (1-f)^2\sin^2\phi] \tag{10b}$$

where f is the flattening parameter of the reference ellipsoid, in this case

$$f = 1/298.257 \tag{11}$$

Thus equation (8) with equations (10), (11) can be used to estimate δr for any geographical latitude ϕ and longitude λ.

Computational scheme

It is evident from equation (8) that computation of the geoid surface is heavily loaded with calculations of the Associated Legendre functions. If the summations in equation (3) are truncated at $\ell = L$, we need at least $L^2/2$ functions to estimate geoid height at each position. An Associated Legendre function of degree n and order m may be defined as:

$$P_{nm}(\cos\theta) = \sin^m\theta \frac{d^m}{d(\cos\theta)^m}[P_{n0}(\cos\theta)] \tag{12}$$

where $$P_{n0}(\cos\theta) = P_n(\cos\theta) = \frac{1}{2^n n!}\frac{d^n}{d(\cos\theta)^n}(\cos^2\theta-1)^n \tag{13}$$

is the Legendre Polynomial of degree n.

Only functions of low degree can easily be derived by expanding the above expressions; expansions become very long and numerically unstable for functions of high degree, and the use of recurrence relationships becomes a necessity. A combination of suitable recurrence relationships is:

$$P_{n+1\,n+1}(\cos\theta) = (2n+1)\sin\theta\, P_{nn}(\cos\theta) \tag{14A}$$

$$P_{n+1\,n}(\cos\theta) = (2n+1)\cos\theta\, P_{nn}(\cos\theta) \tag{14B}$$

$$(n-m+1)P_{n+1\,m}(\cos\theta) = (2n+1)\cos\theta\, P_{nm}(\cos\theta) - (n+m)P_{n-1\,m}(\cos\theta) \tag{14C}$$

If a function P_{nm} is stored as an element of a matrix, the appropriate order of evaluating these functions is shown in Figure 1.

The main advantages of using these recurrence relations are:

(a) they are simple and economical to use.
(b) they are always convergent, and
(c) it is easy to take care of the normalising factor N for each function because only neighbouring functions are used.

The order of processing functions, shown in Figure 1, suggests that a function of degree n and order m (m<n) requires only n intermediate functions. Whereas other relations need all functions of degree less than n and order less than m. This aspect may prove very beneficial in jobs where only a single function is used.

Conclusions

The computer program is tested with tesseral harmonic coefficients from the GEM 10B model, as tabulated in Lerch et al. (1981). The GEM 10B field is complete in harmonics to degree 36 and dimensions in the program are adjusted accordingly. The geoid surface for $0 \leq \lambda \leq 180^\circ$, computed using the above method, as shown in Figure 2, agrees very well with published diagrams of the GEM 10B Geoid. A high resolution geoid covering the region of Ireland and Great Britain is also computed, see Figure 4. Three dimensional projections of these surfaces are obtained (Figures 3,5) to get a clear picture of variations in the geoid.

Computer program

```
      SUBROUTINE GEOID( NLG, GLAT,GLONG,C,S, ODR)
      IMPLICIT REAL*8(A-H,P-Z)
      DIMENSION P(37,37), PR(37,37)
      DIMENSION C(37,37), S(37,37)
      DIMENSION ERD(40), RSUM(40), GLONG(1)
      DIMENSION ODR(1)
      DATA ISKIP/1/
C *****************************************************
C ***** INPUTS TO THE SUBROUTINE  ****                *
C  1 NLG , NUMBER OF POINTS ON THE SAME LATITUDE      *
C        FOR WHICH GEOID IS TO BE CALCULATED.         *
C  2 GLAT, LATITUDE  OF  POINTS                       *
C                                                     *
C  3 GLONG, AN ARRAY WITH LONGITUDES (=NLG) CORRESPONDING
C         TO EACH POINT ON THE ABOVE LATITUDE (GLAT).*
C  4 C AND S, NORMALISED TESSERAL HARMONIC COEFFICIENTS*
C     IN THIS PROGRAM ARRAYS C(I,J), S(I,J), P(I,J)   *
C     CORRESPOND TO C(I-1,J-1), S(I-1,J-1), P(I-1,J-1) *
C     REPECTIVELY IN THE TEXT (EQUATION 1).           *
C     SET     C(1,1),C(2,1),C(2,2),S(1,1),S(2,1),S(2,2)*
C     TO ZERO HERE.                                   *
C ****** OUTPUTS   ******                             *
```

```
C  1 ODR    AN ARRAY RETURNS WITH GEOID VALUES (=NLG)  *
C           CORRESPONDING TO  GLAT, GLONG(I), I=1,NLG  *
C  2        LATITUDE , LONGITUGE AND GEOID ARE PRINTED *
C ****************************************************
      IPR = 6
      ICR = 5
      N =37
      M =37
      N1 = N-1
      M1 = M-1
      RAD = 3.14159265359D0/180.0D0
      IF (ISKIP .EQ. 0) GO TO 9
      CALL LGNDR3 (N,M,0.0D0, ROH,TH,P)
      PSUM = 0.0D0
      DO 8 I = 3, N
 8    PSUM = PSUM + C(I,1)*P(I,1)
      ALPH = 0.00173070D0
      VDZ = 1.0D0 + ALPH + PSUM
      ISKIP = 0
       WRITE (IPR,150) VDZ
 150   FORMAT ('1   VDZ =', F14.10)
C      VDZ = 1.0022728546D0 FROM GEM10B .
 9    CONTINUE
      CALL LGNDR3(N,M,GLAT,ROH,TH, P)
      THR = TH*RAD
      X = DCOS(THR)
      Y = DSIN(THR)
      RHS= ROH*ROH
      AA = 3.0D0*P(3,1)*C(3,1)/RHS
      DVDR = 2.0D0*ROH*ALPH*Y*Y-(1.D0+AA)/RHS
      CALL RDIUSP (ROH,ERD)
C
      DO 20 KK = 1, NLG
      GLKK = GLONG(KK)
      CALL COEFF(M,N,GLKK,C,S,RSUM,P,PR )
C
      SUMB =1.0D0
      DO 19 I = 3, N
 19   SUMB = SUMB +  RSUM(I)/ERD(I)
      CC = ALPH*RHS*Y*Y
      VD = CC+ SUMB/ROH
      ODR(KK) = -6378.138D03*(VD-VDZ)/DVDR
      WRITE (IPR,650) GLAT,GLKK,ROH,VD,DVDR,ODR(KK)
 650  FORMAT ('0 LAT=',F9.4,' LONG=',F9.4,' ROH=',F12.8,' VD=',F12.8,
     & ' DVDR=',F12.8,' DR=',F9.2)
 20   CONTINUE
      RETURN
      END
       SUBROUTINE COEFF (M,N,FLEM,C,S,RSUM,P,PR)
       IMPLICIT REAL*8(A-H,P-Z)
       DIMENSION P(N,M), PR(N,M),RSUM(N)
       DIMENSION C(37,37),S(37,37)
       RAD = 3.14159265359D0/180.0D0
C
```

```
      PSUM = 0.0D0
      J = 1
      DO 20 I = 3,N
      PR(I,J) = C(I,J)*P(I,J)
 20   CONTINUE
C
      J = 2
      JS = 3
C
 26   CONTINUE
      AJ = J-1
      ANG = AJ*FLEM
      JT = ANG/360.0D0
      ANG =(ANG- JT*360.0D0)*RAD
      ANC = DCOS (ANG)
      ANS = DSIN(ANG)
      DO 30 I = JS,N
      PR(I,J) = (C(I,J)*ANC + S(I,J)*ANS )*P(I,J)
 30   CONTINUE
C
      J = J +1
      JS = J
      IF ( J-M) 26,26, 44
C
 44   CONTINUE
      IR =3
 46   RSUM(IR) = 0.0D0
      DO 48 J =1, IR
 48   RSUM(IR)= RSUM(IR) + PR(IR,J)
      IR = IR +1
      IF(IR -N)46,46,50
 50   CONTINUE
C
      RETURN
      END
      SUBROUTINE LGNDR3 (N,M,GLAT,ROH, TH,P )
      IMPLICIT REAL*8(A-H,P-Z)
      DIMENSION P(N,M)
C   COMPUTATION OF  NORMALISED FUNCTIONS  P(N,M)*N(N,M).
C   FIRST A FEW FUNCTIONS ARE GENERATED BY THE DEFINITION.
C   THEN IT COMPUTES ALL FUNCTIONS FOR N=M.
C   AFTER THAT ALL THOSE FOR N= M, M+1...36 FOR M =0,1...35.
C   ARE GENERATED.
      RAD = 3.14159265359D0/180.D0
      CALL ROHNTH (GLAT,ROH,TH )
      THR = TH*RAD
      X = DCOS(THR)
      P(1,1) = 1.0D0
      P(2,1) = X
      P(3,1) = 1.50D0*X*X -0.50D0
      P(4,1) = ((5.00D0*X*X-3.00D0)*X)/2.00D0
      P(5,1) = (7.00D0*X*P(4,1) -3.00D0*P(3,1) )/4.00D0
      AN = DSIN(THR)
      ANB =AN*AN
```

```
      ANC = ANB*ANB
      P(2,2) = AN
      P(3,2) = 3.00D0*X*AN
      P(4,2) = 1.50D0*(5.00D0*X*X -1.0D0)*AN
      P(5,2) = 2.50D0*(7.00D0*X*X-3.00D0)*X*AN
      P(3,3) = 3.00D0*ANB
      P(4,3) = 15.00D0*X*ANB
      P(5,3) = 7.50D0*ANB*(7.00D0*X*X-1.0D0)
      P(4,4) = 15.00D0*AN*ANB
      P(5,4) = 105.0D0*X*AN*ANB
      P(5,5) = 105.0D0*ANC
C
C
      DO 44 J = 1,5
      JA = J-1
      DO 42 I = J,5
      IA = I-1
      FCTR =1.0D0
      IF (JA) 38,38,34
 34   KP = 2*JA
      IJS= IA+ JA
      KIM =1
      DO 36 KK = 1, KP
      KIM  = KIM*IJS
      IJS= IJS-1
 36   CONTINUE
      FCTR =  KIM
      FCTR =DSQRT(FCTR)
 38   CONTINUE
      P(I,J) = P(I,J)/FCTR
 42   CONTINUE
 44   CONTINUE
      IF (N .LE. 5) GO TO 80
      DO 52 J = 6, M
      J1 = J-1
      CC = 2*J1-1
      AA = (2*J1-1)*(2*J1-2)
      BB = 2*J1*(2*J1-1)
      P(J,J1)= (CC*AN*P(J1,J1-1))/DSQRT(AA)
      P(J,J) = (CC*AN*P(J1,J1))/DSQRT(BB)
 52   CONTINUE
C
      M2 = M-2
      DO 60 J = 1,M2
      JA = J-1
      IL = J+2
      IF (IL .LT. 6) IL = 6
      DO 56 I =IL,N
      IA = I-1
      AA = (2*IA-1)*X*P(I-1,J)
      BB = (IA+JA-1)*P(I-2,J)
      CC = IA-JA
      DD=1.0D0
      EE = DD
```

```
      IF (J .EQ. 1) GO TO 54
      AJ = I-J
      BJ = I+J-2
      DD = DSQRT(AJ/BJ)
      AJ = (AJ*(AJ-1.0D0))/(BJ*(BJ-1.0D0))
      EE = DSQRT(AJ)
 54   P(I,J) = ( AA*DD -BB*EE)/CC
 56   CONTINUE
 60   CONTINUE
C
 80   CONTINUE
C
      DO 86 I = 2,N
      AI = 2*(I-1) +1
      P(I,1) = DSQRT(AI)*P(I,1)
 86   CONTINUE
      IR = 2
 92   CONTINUE
      AI = 2*(2*(IR-1) +1)
      AI = DSQRT(AI)
      DO 96 J =2, IR
 96   P(IR,J) = AI*P(IR,J)
      IR = IR +1
      IF( IR-N) 92,92,98
 98   CONTINUE
      RETURN
      END
     SUBROUTINE ROHNTH (GLAT,ROH,TH)
     IMPLICIT REAL*8 (A-H,P-Z)
C    COMPUTES ROH (=R/A) AND CO-LATITUDE (TH)
C    FOR A GIVEN LATITUDE (GLAT).
     RAD= 3.14159265359D0/180.0D0
     THA = GLAT*RAD
     F = 1.0D0/298.257D0
     PF = 1.0D0-F
     PF = PF*PF
     CY = DCOS(THA)
     SY = DSIN(THA)
     R = DSQRT(CY*CY + PF*SY*SY)
     X = CY/R
     Z = PF*SY/R
     ROH = DSQRT(X*X + Z*Z)
     CT = Z/ROH
     ST = X/ROH
     X = ST/CT
     TH = DATAN(X)
     TH= TH/RAD
     IF (X .LT. 0.0D0 ) TH = TH+180.0D0
     IF (ST .LT. 0.0D0) TH = TH+180.0D0
  30 RETURN
     END
       SUBROUTINE RDIUSP (ROH,AR)
       IMPLICIT REAL*8(A-H,P-Z)
       DIMENSION AR(40)
```

```
C       COMPUTES (R/A)**L OR ROH**L (L=0,1..40), VALUES ARE
C       STORED IN AN ARRAY AR.
         BB = ROH*ROH
         AR(1) = 1.0D0
         AR(2) = ROH
         AR(3) = BB
         AR(4) = ROH*BB
         AR(5) = BB*BB
         AR(6) = BB*AR(4)
         AR(7) =BB*AR(5)
         AR(8) = AR(4)*AR(5)
         AR(9) = AR(5)*AR(5)
         AR(10) = ROH*AR(9)
         DO 10 J =11, 20
         AR(J) = AR(J-9) *AR(10)
         AR(J+10)=AR(J)*AR(11)
         AR(J+20)=AR(J+10)*AR(11)
 10      CONTINUE
C        WRITE (6,200) AR
  200    FORMAT (5X,F12.8)
         RETURN
         END
```

References

LERCH, F.J., PUTNEY, B.H., WAGNER, C.A. & KLOSKO, S.M. 1981. Goddard Earth models for oceanographic applications (GEM 10B and 10C). Marine Geodesy, 5, 145-187.

TORGE, W. 1980. Geodesy, an introduction (Translation of "Geodäsie"). Berlin: Water de Gruyter. 254pp.

WUNSCH, C. & GASPOSCHKIN E. M. 1980. On using satellite altimetry to determine the general circulation of the oceans with application to geoid improvement. Reviews of Geophysics and Space Physics, 18, 725-745.

Figure captions

Figure 1. Order of computing Associated Legendre functions: I, initialised by expanding Equation (12); A, B and C indicate recurrence relations in Equation (14) used to derive function P . Pecked line shows the order to compute function P .

Figure 2. Geoid surface for $0 \leqslant \lambda \leqslant 180$ constructed by using tesseral harmonic coefficients from GEM10B model. Heights are in metres above the mean ellipsoid. Contour interval is 5 metres.

Figure 3. Three dimensional projection of the geoid surface in Figure 2

Figure 4. High resolution (0.5 metres) geoid surface around Great Britain and Ireland.

Figure 5. Three dimensional projection of the geoid surface in Figure 4


m → 0 1 2 ... j 36
n ↓
0 I
1 I I
2 I I I
3 C C B A
4 C C C B A
5 . . . . B A
P$_{ij}$
. . . . . C B A
. . . . . C . B A
i . . . . C . . B A
36 C C . . . . B A

Figure 1.


Latitude
Longitude
0
30
60
90
120
150
180
80
60
40
20
-20
-40
-60
-80
-25
25
50
75
100

Figure 2.



Figure 3.



-20
-10
0
70
60
50
40
-10
0
10
Latitude
Longitude
55
60
50
40
45
60
55
50

Figure 4.



Figure 5